Név:

1. dolgozat
Számítógéppel segített matematikai modellezés
"M" változat
2012. október 30, kedd

Oldd meg a következő feladatokat. Készíts szép notebook-ot, figyelj a korrekt strukturált megoldásokra, válaszolj a kérdésekre.

---

1. feladat (12 pont)

a) Készítsd el az alábbi táblázat elsô 20 sorát!

a j-edik sorban a j-edik természetes szám áll, utána a pozitív osztóinak listája, majd a pozitív osztóinak a száma, tehát pl. a 10. sor:

```
10    1 2 5 10    4
```

b) Add meg a tökéletes számokat 100000-ig, azaz olyan n természetes számokat generálj, melyre ((n osztóinak összege = 2n) ∧ (n<10^5)). Pl. (2*6=1+2+3+6)

c) Add meg az $f(x) = \ln(1+x)$ fgv $f^{(k)}$ magasabbrendû deriváltjait táblázatban(0≤k≤10) . Adj ez alapján képletet $f^{(n)}(x)$-re! (n∈ℕ)

a)

```
Table[{j, {Divisors[j]}, Length[Divisors[j]]}, {j, 20}] // TableForm
```

1	1	1
2	1 2	2
3	1 3	2
4	1 2 4	3
5	1 5	2
6	1 2 3 6	4
7	1 7	2
8	1 2 4 8	4
9	1 3 9	3
10	1 2 5 10	4
11	1 11	2
12	1 2 3 4 6 12	6
13	1 13	2
14	1 2 7 14	4
15	1 3 5 15	4
16	1 2 4 8 16	5
17	1 17	2
18	1 2 3 6 9 18	6
19	1 19	2
20	1 2 4 5 10 20	6

b) (A második a OEIS alapján)

```
Select[Table[j, {j, 100000}], Plus @@ Divisors[#] == 2 # &]
```

{6, 28, 496, 8128}

```
(# (# + 1)) / 2 & /@ Select[2^Range[10] - 1, PrimeQ]
```

{6, 28, 496, 8128}

c)

```
f[x_] := Log[1 + x]
```

```
Table[{j, D[f[x], {x, j}]}, {j, 0, 10}] // TableForm
```

0	Log[1 + x]
1	$\frac{1}{1+x}$
2	$-\frac{1}{(1+x)^2}$
3	$\frac{2}{(1+x)^3}$
4	$-\frac{6}{(1+x)^4}$
5	$\frac{24}{(1+x)^5}$
6	$-\frac{120}{(1+x)^6}$
7	$\frac{720}{(1+x)^7}$
8	$-\frac{5040}{(1+x)^8}$
9	$\frac{40\,320}{(1+x)^9}$
10	$-\frac{362\,880}{(1+x)^{10}}$

c) válasz: $f^n(x) = (-1)^{n+1}(1+x)^{-n}\ (n-1)!\,(n>0)$

2. feladat (10 pont)

Adott az alábbi PP_n $(n > 1)$ polinomsorozat.

a) Statikusan ábrázold a gyököket a komplex számsíkon (n=2,...,15)
b) Statikusan ábrázold a gyököket a komplex számsíkon (n=2,...,15) az index szerint színezve
c) Dinamikusan ábrázold a gyököket a komplex számsíkon (n=2,...,15) az index szerint.

```
In[6]:= Clear[P]
```

```
In[24]:= PP[n_] := 1/2^(n-1) (ChebyshevT[n, x] + 2 I ChebyshevT[n-1, x] - ChebyshevT[n-2, x])
```

a)

```
In[25]:= ZList = Table[x /. NSolve[PP[n] == 0], {n, 2, 15}];
```

```
In[14]:= Show[Graphics[{Red, Point[Flatten[ZList /. {x_?NumberQ → {Re[x], Im[x]}}, 1]]}],
 PlotRange → {{-1, 1}, {-1, 0}}]
```

Out[14]=



b)

```
In[26]:= ZList = Table[x /. NSolve[PP[n] == 0], {n, 2, 15}];
```

```
In[16]:= Show[
 Graphics[MapIndexed[# /. {x_?NumberQ → {Hue[#2[[1]] / 16], Point[{Re[x], Im[x]}]}} &, ZList]],
 PlotRange → {{-1, 1}, {-1, 0}}]
```

Out[16]=



c)

```
In[27]:= ZList = Table[x /. NSolve[PP[n] == 0], {n, 2, 15}];
```

In[23]:=
```
Animate[
 Show[Graphics[{Red, PointSize[.02], Point[ZList[[n - 1]] /. {x_?NumberQ → {Re[x], Im[x]}}]}],
  PlotRange → {{-1, 1}, {-1, 0}}, Axes → True, AxesLabel → {"Re", "Im"}], {n, 2, 15, 1}]
```

Out[23]=



3. feladat (10 pont)

Ábrázold a

$$h(x) = \frac{x}{\sqrt[3]{x^2-1}}$$

függvényt a [-4,4] intervallumon.
Add meg $+\infty$-ben a limeszt!
Hogyan kell az elsô ábrát módosítani, hogy ez vizuálisan is szemléltethetô legyen?
Add meg az inflexiós pontokat, színezz (új ábrán) az második derivált elôjele szerint!

```
h[x_] := x / (x^2 - 1)^(1/3)
```

A fgv def nem jó, mert $D_h = \mathbb{R} \setminus \{-1, 1\}$, komplex aritmetika (!) ⟶pl. h2

```
Plot[h[x], {x, -4, 4}]
```



$$\text{h2}[\text{x\_}] := \text{Piecewise}\left[\left\{\left\{\frac{x}{\sqrt[3]{x^2-1}}, x\string^2>1\right\}, \left\{\frac{-x}{\sqrt[3]{1-x^2}}, \text{True}\right\}\right\}\right]$$

```
Plot[h2[x], {x, -4, 4}]
```




```
Limit[h2[x], x → ∞]
```

∞

```
Plot[h2[x], {x, 0, 20}]
```



```
iplist = x /. Solve[D[h2[x], {x, 2}] == 0, x]
```

```
{-3, 3, 0}
```

```
Plot[h2[x], {x, -4, 4},
 ColorFunction → (If[(Evaluate[D[h2[x], {x, 2}]] /. x → #1) > 0, Red, Blue] &),
 ColorFunctionScaling → False, Epilog → {PointSize[.02], Green, Map[Point[{#, h2[#]}] &, iplist]}]
```

Power::infy : Infinite expression $\frac{1}{0^{1/3}}$ encountered ≫

Power::infy : Infinite expression $\frac{1}{0^{1/3}}$ encountered ≫



4. feladat (10 pont)

Forgass meg a P={1,2} pont körül egy

a) pontot
b) egy egységoldalú szabályos háromszöget!

Az interaktív ábrán (csak) az elforgatás α szögét lehessen egy csúszkával állítani.

In[1]:= RotM[α_] := $\begin{pmatrix} \text{Cos}[\alpha] & -\text{Sin}[\alpha] \\ \text{Sin}[\alpha] & \text{Cos}[\alpha] \end{pmatrix}$

In[2]:= `P = {1, 2};`

In[3]:=
```
Manipulate[Show[Graphics[{Point[P], Blue, Point[p], Red, Point[(RotM[α].(p - P)) + P]}],
  AspectRatio → 1, PlotRange → {{-2, 4}, {-2, 4}}],
 {α, -π, π, .1, Appearance → "Labeled"}, {p, {0, 0}, {1, 1}, Locator}]
```

Out[3]=



In[3]:=
```
Tri = {{2, 2}, {3, 2}, {5/2, Sqrt[3]/2 + 2}};
```

```
Manipulate[Show[Graphics[{Point[P], Blue, Polygon[Tri], Red,
    Polygon[Tri /. {x_?NumericQ, y_?NumericQ} -> RotM[α].({x, y} - P) + P]}],
  AspectRatio → 1, PlotRange → {{-2, 5}, {-2, 5}}], {α, -π, π, .1, Appearance → "Labeled"}]
```



4. feladat (8 pont)

A. Legyen $L[x] = r\ x\ (1 - x)$

Add meg a következô iteráltsorozat elsô 8 elemét:
r=1, x0=1/4. Milyen tulajdonságát lehet megsejteni az iterált-sorozatnak?

Add meg a fixpontokat, ha r=2!

Konstruálj meg egy 3-ciklust, ha r=4. (Tipp: Reduce)



```
L[r_, x_] := r x (1 - x)
```

```
L[1, x]
```

```
(1 - x) x
```

```
NestList[L[1, #] &, 1 / 4, 8] // N
```

```
{0.25, 0.1875, 0.152344, 0.129135, 0.112459, 0.0998122, 0.0898497, 0.0817767, 0.0750893}
```

```
x /. Solve[L[2, x] == x, x]
```

$$\left\{0, \frac{1}{2}\right\}$$

```
Plot[{L[2, x], x}, {x, -2, 2}]
```




```
Reduce[L[4, L[4, L[4, x]]] == x ∧ L[4, x] ≠ x ∧
  L[4, L[4, x]] ≠ x ∧ L[4, x] ≠ L[4, L[4, x]], x, Reals]
```

```
x == Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 1] || x == Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 2] ||
 x == Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 3] || x == Root[-3 + 36 #1 - 96 #1^2 + 64 #1^3 &, 1] ||
 x == Root[-3 + 36 #1 - 96 #1^2 + 64 #1^3 &, 2] || x == Root[-3 + 36 #1 - 96 #1^2 + 64 #1^3 &, 3]
```

```
N[%]
```

```
x == 0.188255 || x == 0.61126 || x == 0.950484 || x == 0.116978 || x == 0.413176 || x == 0.969846
```

```
NestList[L[4, #] &, Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 1], 9] // RootReduce // TableForm
```

```
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 1]
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 2]
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 3]
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 1]
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 2]
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 3]
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 1]
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 2]
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 3]
Root[-7 + 56 #1 - 112 #1^2 + 64 #1^3 &, 1]
```

```
N[%]
```

```
{0.188255, 0.61126, 0.950484, 0.188255, 0.61126, 0.950484, 0.188255, 0.61126, 0.950484, 0.188255}
```